Seminar on Computer Algebra
Faculty of Computational Mathematics and Cybernetics of MSU, and Computing Centre of RAS
September 21, 2022, Moscow, Russia

Resonances and Periodic Motion of Atwood's Machine with Two Oscillating Bodies

Alexander N. Prokopenya

Warsaw University of Life Sciences – SGGW, Poland

E-mail: alexander_prokopenya@sggw.pl

Introduction

An Atwood's machine is a well-known device that is usually used in the course in physics to demonstrate the uniformly accelerated motion of a system. It consists of two bodies, having masses m_1, m_2 ($m_2 \geq m_1$), attached to opposite ends of a massless inextensible thread wound round a massless, frictionless pulley.



Using Newton's second law, one can easily obtain its equation of motion and show that acceleration of each body is given by

$$a = \frac{m_2 - m_1}{m_2 + m_1} g,$$

where g is a gravitational acceleration.

Model Description

An Atwood's machine under consideration is shown below. Its geometrical configuration can be described in terms of three variables, namely, a length r of the thread between the body m_1 and the point where the thread departs from the pulley, and the angles φ, ψ determining deviations of the thread from a vertical.



One can readily see that Cartesian coordinates x_1, x_2, y_1, y_2 of the bodies can be determined as

In[ ]:=
```
x1[t_] = r[t] × Sin[φ[t]];
y1[t_] = -r[t] × Cos[φ[t]];
x2[t_] = -b + (L - r[t]) Sin[ψ[t]];
y2[t_] = -(L - r[t]) Cos[ψ[t]];
```

where $r(t)$ and $L - r(t)$ are the lengths of the two pendula, and b is a distance between the pulleys.

Equations of motion

Kinetic energy of the system can be written as

In[ ]:=
```
Ekin = m1/2 (D[x1[t], t]^2 + D[y1[t], t]^2) + m2/2 (D[x2[t], t]^2 + D[y2[t], t]^2) // Collect[#, {r'[t], φ1'[t], φ2'[t], g, m1, m2}, Simplify] &
```

Out[ ]=

$$\left(\frac{m_1}{2}+\frac{m_2}{2}\right) r'[t]^2+\frac{1}{2} r[t]^2 m_1 \varphi'[t]^2+\frac{1}{2} (L-r[t])^2 m_2 \psi'[t]^2$$

Potential energy of the system is determined as

In[ ]:=
```
Epot = m2 g y2[t] + m1 g y1[t] // Simplify
```

Out[ ]=

$$-g \operatorname{Cos}[\varphi[t]] \times r[t]\, m_1+g \operatorname{Cos}[\psi[t]]\,(-L+r[t])\, m_2$$

Therefore, the Lagrangian function for the system can be written in the form

In[ ]:=

```
Lag = Ekin - Epot // Collect[#, {r'[t], φ'[t], ψ'[t], g, m1, m2}, Simplify] &
```

Out[ ]=

$$g\,(\mathrm{Cos}[\varphi[t]]\times r[t]\,m_1+\mathrm{Cos}[\psi[t]]\,(L-r[t])\,m_2)+\left(\frac{m_1}{2}+\frac{m_2}{2}\right)r'[t]^2+\frac{1}{2}\,r[t]^2\,m_1\,\varphi'[t]^2+\frac{1}{2}\,(L-r[t])^2\,m_2\,\psi'[t]^2$$

Equations of motion don't change if the Lagrangian is multiplied by a constant factor. Besides, let us introduce parameter $\varepsilon = \frac{m_2-m_1}{m_1}$ and dimensionless length $r^* = r / R_0$ and time $t^* = \sqrt{g / R_0}\ t$, then $L = R_0\, k$.

Therefore, the Lagrangian can be written in the form

In[ ]:=

```
L1 =
 Lag / (m1 g R0) /. L → (R0 k) /. r → (R0 r[#] &) /. {Derivative[k_][f_][t] → (g/R0)^(k/2) Derivative[k][f][t]} /.
   m2 → (1 + ε) m1 // Collect[#, {r'[t], φ1'[t], φ2'[t]}, Simplify] &
```

Out[ ]=

$$\frac{1}{2}\,(2+\varepsilon)\,r'[t]^2+\frac{1}{2}\left(k\,(1+\varepsilon)\,\left(2\,\mathrm{Cos}[\psi[t]]+k\,\psi'[t]^2\right)+r[t]^2\,\left(\varphi'[t]^2+(1+\varepsilon)\,\psi'[t]^2\right)+2\,r[t]\,\left(\mathrm{Cos}[\varphi[t]]-(1+\varepsilon)\,\mathrm{Cos}[\psi[t]]-k\,(1+\varepsilon)\,\psi'[t]^2\right)\right)$$

Then equations of motion written in the Lagrange form are given by

In[ ]:=

```
eq = {D[D[L1, r'[t]], t] == D[L1, r[t]],
     D[D[L1, φ'[t]], t] == D[L1, φ[t]],
   D[D[L1, ψ'[t]], t] == D[L1, ψ[t]] };
eq1 = {r''[t], φ''[t], ψ''[t]} == ({r''[t], φ''[t], ψ''[t]} /. Solve[eq, {r''[t], φ''[t], ψ''[t]}][[1]] //
     Collect[#, {r′[t], φ′[t], ψ'[t]}, Simplify] &) // Thread
```

Out[ ]=

$$\left\{r''[t] == \frac{\mathrm{Cos}[\varphi[t]] - (1+\varepsilon)\,\mathrm{Cos}[\psi[t]]}{2+\varepsilon} + \frac{r[t]\,\varphi'[t]^2}{2+\varepsilon} - \frac{(1+\varepsilon)\,(k-r[t])\,\psi'[t]^2}{2+\varepsilon},\ \varphi''[t] == -\frac{\mathrm{Sin}[\varphi[t]]}{r[t]} - \frac{2\,r'[t]\,\varphi'[t]}{r[t]},\ \psi''[t] == -\frac{\mathrm{Sin}[\psi[t]]}{k-r[t]} + \frac{2\,r'[t]\,\psi'[t]}{k-r[t]}\right\}$$

Equal masses - simulation ($m_1 = m_2$)

Let us consider the case when this system rests and we impart small horizontal initial velocities v_{01} and v_{02} to the bodies. What kind of the system motion can we expect by intuition?



As both bodies have the same mass and are initially in equilibrium, it seems to be quite natural to assume that the system will oscillate near its equilibrium position. However, solving the equations of motion *eq1* shows that the system demonstrates completely different behaviour.

Actually, choosing some realistic values of the system parameters and using the function **NDSolve**, one ca easily find numerical solution of equations *eq1*, satisfying the corresponding initial conditions. At first only one point gets some initial velocity.

In[ ]:=

```
rul1 = {ε → 0, k → 2, dφ1 → 0.3, dφ2 → -0.3};
initial1 = {φ[0] == 0, φ'[0] == dφ1,
          ψ[0] == 0, ψ'[0] == dφ2,
          r[0] == 1, r'[0] == 0}; (* bodies m1 and m2 are pushed in horizontal direction *)
sol1 = NDSolve[Join[eq1, initial1] /. rul1, {φ, ψ, r}, {t, 0, 80}][[1]];
Row[{Plot[{r[t] /.sol1} // Evaluate, {t, 0, 80}, AxesLabel → {"t", "r"}, PlotStyle → {Thick, Red}, BaseStyle → {FontFamily → "Arial", FontSize → 12}, ImageSize → 250, GridLines → Automatic],
  Plot[{φ[t] /.sol1, ψ[t] /.sol1}, {t, 0, 70}, AxesLabel → {"t", "φ,ψ"}, PlotStyle → {{Thick, Green}, {Thick, Blue}},
   BaseStyle → {FontFamily → "Arial", FontSize → 12}, ImageSize → 250, GridLines → Automatic]}]
```

Out[ ]=



In[ ]:=

```
rul1 = {ε → 0, k → 2, dφ1 → 0.3, dφ2 → -0.3};
initial1 = {φ[0] == dφ1, φ'[0] == 0,      (* bodies are deviated from the vertical *)
          ψ[0] == dφ2, ψ'[0] == 0,
          r[0] == 1, r'[0] == 0};
sol1 = NDSolve[Join[eq1, initial1] /. rul1, {φ, ψ, r}, {t, 0, 80}][[1]];
Row[{Plot[{r[t] /.sol1} // Evaluate, {t, 0, 80}, AxesLabel → {"t", "r"}, PlotStyle → {Thick, Red}, BaseStyle → {FontFamily → "Arial", FontSize → 12}, ImageSize → 250, GridLines → Automatic],
  Plot[{φ[t] /.sol1, ψ[t] /.sol1}, {t, 0, 70}, AxesLabel → {"t", "φ,ψ"}, PlotStyle → {{Thick, Green}, {Thick, Blue}},
   BaseStyle → {FontFamily → "Arial", FontSize → 12}, ImageSize → 250, GridLines → Automatic]}]
```

Out[ ]=



If initial lengths of the pendula are different one can choose such initial conditions that the bodies oscillate periodically with different frequencies while the length of the thread oscillates

In[ ]:=

```
rul1 = {ε → 0, k → 5/4, dφ1 → -0.1, dφ2 → 0.1997};
initial1 = {φ[0] == 0, φ'[0] == dφ1,      (* body m1 is pushed in horizontal direction *)
            ψ[0] == 0, ψ'[0] == dφ2,
            r[0] == 1, r'[0] == 0};
sol1 = NDSolve[Join[eq1, initial1] /. rul1, {φ, ψ, r}, {t, 0, 100}][[1]];
Row[{Plot[{r[t] /.sol1} // Evaluate, {t, 0, 80}, AxesLabel → {"t", "r"}, PlotStyle → {Thick, Red}, BaseStyle → {FontFamily → "Arial", FontSize → 12}, ImageSize → 250, GridLines → Automatic],
  Plot[{φ[t] /.sol1, ψ[t] /.sol1}, {t, 0, 50}, AxesLabel → {"t", "φ,ψ"}, PlotStyle → {{Thick, Green}, {Thick, Blue}},
   BaseStyle → {FontFamily → "Arial", FontSize → 12}, ImageSize → 250, GridLines → Automatic]}]
```

Out[ ]=





However, the system demonstrates completely different behaviour if both bodies get different initial velocities.

Periodic motion with two oscillating bodies ($2\,\omega_1 = 3\,\omega_2$)

Let us consider first the case of equal masses of the bodies ($\varepsilon = 0$), then equations of motion are

In[ ]:=

$$eq2 = \left\{r''[t] == \frac{1}{2}(\mathrm{Cos}[\varphi[t]] - \mathrm{Cos}[\psi[t]]) + \frac{1}{2} r[t]\,\varphi'[t]^2 - \frac{1}{2}(k - r[t])\,\psi'[t]^2,\ \varphi''[t] == -\frac{\mathrm{Sin}[\varphi[t]]}{r[t]} - \frac{2 r'[t]\,\varphi'[t]}{r[t]},\ \psi''[t] == -\frac{\mathrm{Sin}[\psi[t]]}{k - r[t]} + \frac{2 r'[t]\,\psi'[t]}{k - r[t]}\right\};$$

Besides, we assume the angles φ, ψ are small and expand trigonometric functions into power series up to seventh order. The equations take the form

In[ ]:=

```
eq2a = r[t] (eq2[[2, 1]] - eq2[[2, 2]]) /. Sin[x_] → (Series[Sin[x], {x, 0, 7}] // Normal) // Simplify
```

Out[ ]=

$$\varphi[t] - \frac{\varphi[t]^3}{6} + \frac{\varphi[t]^5}{120} - \frac{\varphi[t]^7}{5040} + 2 r'[t]\,\varphi'[t] + r[t]\,\varphi''[t]$$

In[ ]:=

```
eq2b = (k - r[t]) (eq2[[3, 1]] - eq2[[3, 2]]) /. Sin[x_] → (Series[Sin[x], {x, 0, 7}] // Normal) // Simplify
```

Out[ ]=

$$\psi[t] - \frac{\psi[t]^3}{6} + \frac{\psi[t]^5}{120} - \frac{\psi[t]^7}{5040} - 2\, r'[t]\, \psi'[t] + (k - r[t])\, \psi''[t]$$

In[ ]:=

```
eq2c = (eq2[[1, 1]] - eq2[[1, 2]]) /. Cos[x_] → (Series[Cos[x], {x, 0, 7}] // Normal) // Simplify
```

Out[ ]=

$$\frac{1}{1440} \left(360\, \varphi[t]^2 - 30\, \varphi[t]^4 + \varphi[t]^6 - 360\, \psi[t]^2 + 30\, \psi[t]^4 - \psi[t]^6 + 720 \left(k\, \psi'[t]^2 - r[t] \left(\varphi'[t]^2 + \psi'[t]^2\right) + 2\, r''[t]\right)\right)$$

However, there may exist a state of dynamical equilibrium when the distance oscillates near some equilibrium value.
First, let us introduce dimensionless variables according to the rule

In[ ]:=

```
rul1 = {r → ((1 + ε r[#]) &), φ → (√ε φ[#] &), ψ → (√ε ψ[#] &)};
```

We assume that the system has an equilibrium state $r = 1$, $\varphi = \psi = 0$ if $\varepsilon = 0$.

In[ ]:=

```
eq3a = 1/√ε eq2a /. rul1 // Collect[#, ε, Simplify] &
```

Out[ ]=

$$\varphi[t] + \frac{1}{120}\, \varepsilon^2\, \varphi[t]^5 - \frac{\varepsilon^3\, \varphi[t]^7}{5040} + \varphi''[t] + \varepsilon \left(-\frac{1}{6}\, \varphi[t]^3 + 2\, r'[t]\, \varphi'[t] + r[t]\, \varphi''[t]\right)$$

In[ ]:=

```
eq3b = 1/√ε eq2b /. rul1 // Collect[#, ε, Simplify] &
```

Out[ ]=

$$\psi[t] + \frac{1}{120}\, \varepsilon^2\, \psi[t]^5 - \frac{\varepsilon^3\, \psi[t]^7}{5040} + (-1 + k)\, \psi''[t] + \varepsilon \left(-\frac{1}{6}\, \psi[t]^3 - 2\, r'[t]\, \psi'[t] - r[t]\, \psi''[t]\right)$$

In[ ]:=

```
eq3c = 1/ε eq2c /. rul1 // Collect[#, ε, Simplify] &
```

Out[ ]=

$$\frac{\varepsilon^2 \left(\varphi[t]^6 - \psi[t]^6\right)}{1440} + \frac{1}{48}\, \varepsilon \left(-\varphi[t]^4 + \psi[t]^4 - 24\, r[t] \left(\varphi'[t]^2 + \psi'[t]^2\right)\right) + \frac{1}{4} \left(\varphi[t]^2 - \psi[t]^2 - 2 \left(\varphi'[t]^2 + \psi'[t]^2 - k\, \psi'[t]^2 - 2\, r''[t]\right)\right)$$

For small values of ε we can expand the equations into power series in ε and look for the solutions in the form

In[ ]:=

```
rul2 = {r → ((r0[#] + ε r1[#] + ε^2 r2[#] + ε^3 r3[#]) &), φ → (φ0[#] + ε φ1[#] + ε^2 φ2[#] + ε^3 φ3[#] &), ψ → (ψ0[#] + ε ψ1[#] + ε^2 ψ2[#] + ε^3 ψ3[#] &)};
```

In[ ]:=

```
rulDer = {Derivative[n_][φ][t] → ω1^n Derivative[n][φ][t], Derivative[n_][ψ][t] → (ω1)^n Derivative[n][ψ][t]};
```

In[ ]:=

```
rulω = {ω1 → 1 + ω11 ε + ω12 ε^2 + ω13 ε^3};
```

In[ ]:=

```
eq4a = eq3a /. rulDer /. rul2 /. rulω // Series[#, {ε, 0, 3}] & // Normal // Collect[#, ε, Simplify] &
```

Out[ ]=

$$\varphi0[t] + \varphi0''[t] + \varepsilon\left(-\frac{1}{6}\varphi0[t]^3 + \varphi1[t] + 2\,r0'[t]\,\varphi0'[t] + 2\,\omega11\,\varphi0''[t] + r0[t]\,\varphi0''[t] + \varphi1''[t]\right) +$$
$$\varepsilon^2\left(\frac{\varphi0[t]^5}{120} - \frac{1}{2}\varphi0[t]^2\varphi1[t] + \varphi2[t] + 2\,(\omega11\,r0'[t] + r1'[t])\,\varphi0'[t] + 2\,r0'[t]\,\varphi1'[t] + \left(\omega11^2 + 2\,\omega12\right)\varphi0''[t] + (2\,\omega11\,r0[t] + r1[t])\,\varphi0''[t] + 2\,\omega11\,\varphi1''[t] + r0[t]\,\varphi1''[t] + \varphi2''[t]\right) +$$
$$\varepsilon^3\Bigg(-\frac{\varphi0[t]^7}{5040} + \frac{1}{24}\varphi0[t]^4\varphi1[t] - \frac{1}{2}\varphi0[t]\left(\varphi1[t]^2 + \varphi0[t]\times\varphi2[t]\right) + \varphi3[t] + 2\,(r2'[t]\,\varphi0'[t] + r1'[t]\,(\omega11\,\varphi0'[t] + \varphi1'[t]) + r0'[t]\,(\omega12\,\varphi0'[t] + \omega11\,\varphi1'[t] + \varphi2'[t])) +$$
$$2\,(\omega11\,\omega12 + \omega13)\,\varphi0''[t] + \left(\left(\omega11^2 + 2\,\omega12\right) r0[t] + 2\,\omega11\,r1[t] + r2[t]\right)\varphi0''[t] + \left(\omega11^2 + 2\,\omega12\right)\varphi1''[t] + (2\,\omega11\,r0[t] + r1[t])\,\varphi1''[t] + 2\,\omega11\,\varphi2''[t] + r0[t]\,\varphi2''[t] + \varphi3''[t]\Bigg)$$

In[ ]:=

```
eq4b = eq3b /. rulDer /. rul2 /. k → 13 / 4 /. rulω // Series[#, {ε, 0, 3}] & // Normal // Collect[#, ε, Simplify] &
```

Out[ ]=

$$\psi0[t] + \frac{9\,\psi0''[t]}{4} + \varepsilon\left(-\frac{1}{6}\psi0[t]^3 + \psi1[t] - 2\,r0'[t]\,\psi0'[t] - r0[t]\,\psi0''[t] + \frac{9}{4}(2\,\omega11\,\psi0''[t] + \psi1''[t])\right) +$$
$$\varepsilon^2\left(\frac{\psi0[t]^5}{120} - \frac{1}{2}\psi0[t]^2\psi1[t] + \psi2[t] - 2\,(r1'[t]\,\psi0'[t] + r0'[t]\,(\omega11\,\psi0'[t] + \psi1'[t])) - (2\,\omega11\,r0[t] + r1[t])\,\psi0''[t] - r0[t]\,\psi1''[t] + \frac{9}{4}\left(\left(\omega11^2 + 2\,\omega12\right)\psi0''[t] + 2\,\omega11\,\psi1''[t] + \psi2''[t]\right)\right) +$$
$$\varepsilon^3\Bigg(-\frac{\psi0[t]^7}{5040} + \frac{1}{24}\psi0[t]^4\psi1[t] - \frac{1}{2}\psi0[t]\left(\psi1[t]^2 + \psi0[t]\times\psi2[t]\right) + \psi3[t] - 2\,(r2'[t]\,\psi0'[t] + r1'[t]\,(\omega11\,\psi0'[t] + \psi1'[t]) + r0'[t]\,(\omega12\,\psi0'[t] + \omega11\,\psi1'[t] + \psi2'[t])) -$$
$$\left(\left(\omega11^2 + 2\,\omega12\right) r0[t] + 2\,\omega11\,r1[t] + r2[t]\right)\psi0''[t] - (2\,\omega11\,r0[t] + r1[t])\,\psi1''[t] - r0[t]\,\psi2''[t] + \frac{9}{4}\left(2\,(\omega11\,\omega12 + \omega13)\,\psi0''[t] + \left(\omega11^2 + 2\,\omega12\right)\psi1''[t] + 2\,\omega11\,\psi2''[t] + \psi3''[t]\right)\Bigg)$$

In[ ]:=

```
eq4c = eq3c /. rulDer /. rul2  /. k → 13 / 4 /. rulω // Series[#, {ε, 0, 3}] & // Normal // Collect[#, ε, Simplify] &
```

Out[ ]=

$$\frac{1}{8}\left(2\,\varphi0[t]^2-2\,\psi0[t]^2-4\,\varphi0'[t]^2+9\,\psi0'[t]^2+8\,r0''[t]\right)+$$

$$\frac{1}{48}\,\varepsilon\left(-\varphi0[t]^4+24\,\varphi0[t]\times\varphi1[t]+\psi0[t]^4-24\,\psi0[t]\times\psi1[t]-12\left(2\,(2\,\omega11+r0[t])\,\varphi0'[t]^2+4\,\varphi0'[t]\,\varphi1'[t]+(-9\,\omega11+2\,r0[t])\,\psi0'[t]^2-9\,\psi0'[t]\,\psi1'[t]-4\,r1''[t]\right)\right)+$$

$$\frac{1}{1440}\,\varepsilon^2\left(\varphi0[t]^6-\psi0[t]^6+30\left(-4\,\varphi0[t]^3\,\varphi1[t]+4\,\psi0[t]^3\,\psi1[t]-24\,r1[t]\left(\varphi0'[t]^2+\psi0'[t]^2\right)-48\,r0[t]\left(\omega11\,\varphi0'[t]^2+\varphi0'[t]\,\varphi1'[t]+\psi0'[t]\,(\omega11\,\psi0'[t]+\psi1'[t])\right)\right)+360\left(\varphi1[t]^2+2\,\varphi0[t]\times\varphi2[t]-\psi1[t]^2-\right.\right.$$

$$\left.\left.2\,\psi0[t]\times\psi2[t]-2\left(\left(\omega11^2+2\,\omega12\right)\varphi0'[t]^2+4\,\omega11\,\varphi0'[t]\,\varphi1'[t]+\varphi1'[t]^2+2\,\varphi0'[t]\,\varphi2'[t]-\frac{9}{4}\left(\left(\omega11^2+2\,\omega12\right)\psi0'[t]^2+\psi1'[t]^2+2\,\psi0'[t]\,(2\,\omega11\,\psi1'[t]+\psi2'[t])\right)-2\,r2''[t]\right)\right)\right)+$$

$$\frac{1}{240}\,\varepsilon^3\left(\varphi0[t]^5\,\varphi1[t]-\psi0[t]^5\,\psi1[t]+5\left(-2\,\varphi0[t]^2\left(3\,\varphi1[t]^2+2\,\varphi0[t]\times\varphi2[t]\right)+2\,\psi0[t]^2\left(3\,\psi1[t]^2+2\,\psi0[t]\times\psi2[t]\right)-24\left(r2[t]\left(\varphi0'[t]^2+\psi0'[t]^2\right)+2\,r1[t]\left(\omega11\,\varphi0'[t]^2+\varphi0'[t]\,\varphi1'[t]+\right.\right.\right.\right.$$

$$\left.\left.\psi0'[t]\,(\omega11\,\psi0'[t]+\psi1'[t])\right)+r0[t]\left(\left(\omega11^2+2\,\omega12\right)\varphi0'[t]^2+4\,\omega11\,\varphi0'[t]\,\varphi1'[t]+\varphi1'[t]^2+2\,\varphi0'[t]\,\varphi2'[t]+\left(\omega11^2+2\,\omega12\right)\psi0'[t]^2+4\,\omega11\,\psi0'[t]\,\psi1'[t]+\psi1'[t]^2+2\,\psi0'[t]\,\psi2'[t]\right)\right)\right)+$$

$$60\left(2\,\varphi1[t]\times\varphi2[t]+2\,\varphi0[t]\times\varphi3[t]-2\,\psi1[t]\times\psi2[t]-2\,\psi0[t]\times\psi3[t]-4\,(\omega11\,\omega12+\omega13)\,\varphi0'[t]^2-4\left(\omega11^2+2\,\omega12\right)\varphi0'[t]\,\varphi1'[t]-4\,\varphi1'[t]\,\varphi2'[t]-4\,\omega11\left(\varphi1'[t]^2+2\,\varphi0'[t]\,\varphi2'[t]\right)-\right.$$

$$\left.\left.4\,\varphi0'[t]\,\varphi3'[t]+9\left((\omega11\,\omega12+\omega13)\,\psi0'[t]^2+\psi1'[t]\,(\omega11\,\psi1'[t]+\psi2'[t])+\psi0'[t]\left(\left(\omega11^2+2\,\omega12\right)\psi1'[t]+2\,\omega11\,\psi2'[t]+\psi3'[t]\right)\right)+4\,r3''[t]\right)\right)$$

In zero approximation in ε we obtain

In[ ]:=

```
solφ0 = DSolve[{Coefficient[eq4a, ε, 0] == 0, φ0[0] == 0, φ0'[0] == 1}, φ0, t][[1]]
```

Out[ ]=

```
{φ0 → Function[{t}, Sin[t]]}
```

Function $\psi_0(t)$ is similar but we introduce arbitrary initial phase α.

In[ ]:=

```
solψ0 = {ψ0 → (Sin[2 # / 3] &)};
```

In[ ]:=

```
Coefficient[eq4b, ε, 0] /. solψ0
```

Out[ ]=

```
0
```

On substituting solutions $\varphi_0(t)$, $\psi_0(t)$ into equation for $r_0(t)$, we obtain

In[ ]:=

```
ex4c = (Coefficient[eq4c, ε, 0] /. solφ0 /. solψ0 // TrigReduce)
```

Out[ ]=

$$\frac{1}{8}\left(3\,\mathrm{Cos}\left[\frac{4\,t}{3}\right]-3\,\mathrm{Cos}[2\,t]+8\,r0''[t]\right)$$

Integrating this equation two times over time gives

In[ ]:=

```
solr0 = r0[t] → (Integrate[-ex4c + r0''[t], t, t] + r00)
```

Out[ ]=

$$r0[t] \to r00 + \frac{3}{128}\left(9\,\mathrm{Cos}\left[\frac{4\,t}{3}\right] - 4\,\mathrm{Cos}[2\,t]\right)$$

In the first order in ε we obtain

In[ ]:=

```
eq5a = ((Coefficient[eq4a, ε, 1] /. solφ0 /. solψ0 /. solr0 /. D[solr0, t]) // TrigReduce)  // Expand
```

Out[ ]=

$$-\frac{45}{256}\,\mathrm{Sin}\left[\frac{t}{3}\right] + \frac{\mathrm{Sin}[t]}{64} - r00\,\mathrm{Sin}[t] - 2\,\omega11\,\mathrm{Sin}[t] - \frac{99}{256}\,\mathrm{Sin}\left[\frac{7\,t}{3}\right] + \frac{53}{192}\,\mathrm{Sin}[3\,t] + \varphi1[t] + \varphi1''[t]$$

In[ ]:=

```
eq5b = ((Coefficient[eq4b, ε, 1] /. solφ0 /. solψ0 /. solr0 /. D[solr0, t]) // TrigReduce)  // Expand
```

Out[ ]=

$$\frac{1}{64}\,\mathrm{Sin}\left[\frac{2\,t}{3}\right] + \frac{4}{9}\,r00\,\mathrm{Sin}\left[\frac{2\,t}{3}\right] - 2\,\omega11\,\mathrm{Sin}\left[\frac{2\,t}{3}\right] - \frac{5}{48}\,\mathrm{Sin}\left[\frac{4\,t}{3}\right] + \frac{53}{192}\,\mathrm{Sin}[2\,t] - \frac{7}{48}\,\mathrm{Sin}\left[\frac{8\,t}{3}\right] + \psi1[t] + \frac{9\,\psi1''[t]}{4}$$

In[ ]:=

```
solω11 = Solve[(Coefficient[eq5a , Sin[t]]) == 0, ω11] // First // Simplify
```

Out[ ]=

$$\left\{\omega11 \to \frac{1}{128} - \frac{r00}{2}\right\}$$

In[ ]:=

```
solω11a = Solve[(Coefficient[eq5b , Sin[2 t / 3]]) == 0, ω11] // First // Simplify
```

Out[ ]=

$$\left\{\omega11 \to \frac{1}{128} + \frac{2\,r00}{9}\right\}$$

In[ ]:=

```
solr00 = Solve[(ω11 /. solω11) == (ω11 /. solω11a), r00] // First
```

Out[ ]=

$$\{r00 \to 0\}$$

In[ ]:=

```
(ω11 /. solω11 /. solr00)
```

Out[ ]=

$$\frac{1}{128}$$

In[ ]:=

```
solφ1 = DSolve[{(eq5a /. solω11 /. solr00) == 0, φ1[0] == 0, φ1'[0] == 0}, φ1[t], t][[1]] // Simplify // TrigReduce
```

Out[ ]=

$$\left\{\varphi 1[t] \to \frac{1}{30720}\left(6075\,\text{Sin}\left[\frac{t}{3}\right] + 1032\,\text{Sin}[t] - 2673\,\text{Sin}\left[\frac{7t}{3}\right] + 1060\,\text{Sin}[3t]\right)\right\}$$

In[ ]:=

```
φ1[t] /. solφ1 // Expand
```

Out[ ]=

$$\frac{405\,\text{Sin}\left[\frac{t}{3}\right]}{2048} + \frac{43\,\text{Sin}[t]}{1280} - \frac{891\,\text{Sin}\left[\frac{7t}{3}\right]}{10240} + \frac{53\,\text{Sin}[3t]}{1536}$$

In[ ]:=

```
solψ1 = DSolve[{(eq5b /. solω11 /. solr00) == 0, ψ1[0] == 0}, ψ1[t], t][[1]] /. C[2] → (C2a + 96 / 2304) // Simplify // TrigReduce
```

Out[ ]=

$$\left\{\psi 1[t] \to \frac{1}{23040}\left(23040\,\text{C2a}\,\text{Sin}\left[\frac{2t}{3}\right] - 800\,\text{Sin}\left[\frac{4t}{3}\right] + 795\,\text{Sin}[2t] - 224\,\text{Sin}\left[\frac{8t}{3}\right]\right)\right\}$$

In[ ]:=

```
ψ1[t] /. solψ1 // Expand
```

Out[ ]=

$$\text{C2a}\,\text{Sin}\left[\frac{2t}{3}\right] - \frac{5}{144}\,\text{Sin}\left[\frac{4t}{3}\right] + \frac{53\,\text{Sin}[2t]}{1536} - \frac{7}{720}\,\text{Sin}\left[\frac{8t}{3}\right]$$

In[ ]:=

```
eq5c = ((Coefficient[eq4c, ε, 1] /. solφ0 /. solψ0 /. solr0 /. D[solr0, t] /. solω11 /. solr00 /. solφ1 /. D[solφ1, t] /. solψ1 /. D[solψ1, t]) // TrigReduce) // Expand
```

Out[ ]=

$$-\frac{43}{5120} + \frac{\text{C2a}}{4} - \frac{755\,\text{Cos}\left[\frac{2t}{3}\right]}{24576} - \frac{27}{640}\,\text{Cos}\left[\frac{4t}{3}\right] + \frac{3}{4}\,\text{C2a}\,\text{Cos}\left[\frac{4t}{3}\right] - \frac{181\,\text{Cos}[2t]}{2048} + \frac{105\,\text{Cos}\left[\frac{8t}{3}\right]}{2048} + \frac{9859\,\text{Cos}\left[\frac{10t}{3}\right]}{122880} - \frac{105\,\text{Cos}[4t]}{2048} + r1''[t]$$

In[ ]:=

```
solC2a = Solve[(eq5c /. {f_[k_ t + b_.] → 0, f_''[t] → 0}) == 0, C2a][[1]]
```

Out[ ]=

$$\left\{\text{C2a} \to \frac{43}{1280}\right\}$$

Integrating this equation two times over time gives

In[ ]:=

```
solr1 = r1[t] → (Integrate[-(eq5c /. solC2a) + r1''[t], t, t] + r10)
```

Out[ ]=

$$r1[t] \to r10 + \frac{1}{16384000}\left(-1132500\,\text{Cos}\left[\frac{2t}{3}\right] - 156600\,\text{Cos}\left[\frac{4t}{3}\right] - 362000\,\text{Cos}[2t] + 118125\,\text{Cos}\left[\frac{8t}{3}\right] + 118308\,\text{Cos}\left[\frac{10t}{3}\right] - 52500\,\text{Cos}[4t]\right)$$

Repeating the calculations in higher orders, we obtain finally

In[ ]:=

```
φφ[t_] = √ε φ[t] /. rul2 /. solφ0 /. solφ1 /. solφ2 /. solφ3
```

Out[ ]=

$$\sqrt{\varepsilon}\left(\mathrm{Sin}[t]+\frac{1}{30\,720}\varepsilon\left(6075\,\mathrm{Sin}\left[\frac{t}{3}\right]+1032\,\mathrm{Sin}[t]-2673\,\mathrm{Sin}\left[\frac{7t}{3}\right]+1060\,\mathrm{Sin}[3t]\right)+\frac{1}{5\,242\,880\,000}\varepsilon^2\left(323\,262\,000\,\mathrm{Sin}\left[\frac{t}{3}\right]-481\,827\,650\,\mathrm{Sin}[t]+80\,026\,875\,\mathrm{Sin}\left[\frac{5t}{3}\right]+56\,022\,768\,\mathrm{Sin}\left[\frac{7t}{3}\right]+42\,362\,000\,\mathrm{Sin}[3t]+35\,488\,125\,\mathrm{Sin}\left[\frac{11t}{3}\right]-47\,391\,552\,\mathrm{Sin}\left[\frac{13t}{3}\right]+11\,626\,000\,\mathrm{Sin}[5t]\right)+\frac{1}{12\,153\,683\,705\,856\,000\,000}\varepsilon^3\left(-836\,153\,264\,766\,227\,100\,\mathrm{Sin}\left[\frac{t}{3}\right]-514\,166\,302\,095\,806\,640\,\mathrm{Sin}[t]+104\,528\,328\,238\,950\,000\,\mathrm{Sin}\left[\frac{5t}{3}\right]+433\,721\,435\,074\,537\,764\,\mathrm{Sin}\left[\frac{7t}{3}\right]-141\,595\,864\,927\,247\,500\,\mathrm{Sin}[3t]-3\,564\,853\,970\,898\,000\,\mathrm{Sin}\left[\frac{11t}{3}\right]-4\,814\,285\,409\,050\,271\,\mathrm{Sin}\left[\frac{13t}{3}\right]+4\,192\,398\,803\,862\,500\,\mathrm{Sin}[5t]+18\,492\,698\,446\,778\,160\,\mathrm{Sin}\left[\frac{17t}{3}\right]-11\,825\,907\,007\,224\,975\,\mathrm{Sin}\left[\frac{19t}{3}\right]+2\,073\,940\,484\,000\,000\,\mathrm{Sin}[7t]\right)\right)$$

In[ ]:=

```
ψψ[t_] = √ε ψ[t] /. rul2 /. solψ0 /. solψ1 /. solψ2 /. solψ3 /. solC2a /. solC2b /. solC2c
```

Out[ ]=

$$\sqrt{\varepsilon}\left(\mathrm{Sin}\left[\frac{2t}{3}\right]+\frac{1}{23\,040}\varepsilon\left(774\,\mathrm{Sin}\left[\frac{2t}{3}\right]-800\,\mathrm{Sin}\left[\frac{4t}{3}\right]+795\,\mathrm{Sin}[2t]-224\,\mathrm{Sin}\left[\frac{8t}{3}\right]\right)+\frac{1}{26\,542\,080\,000}\varepsilon^2\left(-\frac{5\,358\,891\,925}{2}\mathrm{Sin}\left[\frac{2t}{3}\right]-576\,372\,000\,\mathrm{Sin}\left[\frac{4t}{3}\right]-4\,039\,875\,\mathrm{Sin}[2t]-79\,872\,672\,\mathrm{Sin}\left[\frac{8t}{3}\right]+62\,994\,125\,\mathrm{Sin}\left[\frac{10t}{3}\right]-13\,313\,952\,\mathrm{Sin}[4t]-638\,750\,\mathrm{Sin}\left[\frac{14t}{3}\right]\right)+\frac{1}{1\,107\,504\,427\,696\,128\,000\,000}\varepsilon^3\left(-58\,839\,620\,906\,636\,990\,910\,\mathrm{Sin}\left[\frac{2t}{3}\right]+5\,507\,452\,608\,789\,331\,200\,\mathrm{Sin}\left[\frac{4t}{3}\right]-12\,998\,246\,977\,108\,008\,000\,\mathrm{Sin}[2t]+590\,185\,625\,099\,169\,792\,\mathrm{Sin}\left[\frac{8t}{3}\right]+440\,127\,684\,161\,550\,000\,\mathrm{Sin}\left[\frac{10t}{3}\right]-524\,137\,388\,880\,745\,728\,\mathrm{Sin}[4t]+270\,764\,533\,684\,346\,625\,\mathrm{Sin}\left[\frac{14t}{3}\right]-98\,458\,072\,864\,940\,800\,\mathrm{Sin}\left[\frac{16t}{3}\right]+24\,154\,523\,186\,107\,545\,\mathrm{Sin}[6t]-3\,035\,357\,920\,000\,000\,\mathrm{Sin}\left[\frac{20t}{3}\right]\right)\right)$$

We can easily visualize the functions obtained.

In[ ]:=

```
Plot[{φφ[t] /. ε → 1/2, ψψ[t] /. ε → 1/2}, {t, 0, 50}]
```

Out[ ]=



In[ ]:=

```
rr[t_] = 1 + ε r[t] /. rul2 /. solr0 /. solr1 /. solr2 /. solr00 /. solr10 /. solr20 /. r3[t] → 0
```

Out[ ]=

$$1+\varepsilon\left(\frac{3}{128}\left(9\,\mathrm{Cos}\left[\frac{4\,t}{3}\right]-4\,\mathrm{Cos}[2\,t]\right)+\varepsilon\left(\frac{5439}{655\,360}+\left(-1\,132\,500\,\mathrm{Cos}\left[\frac{2\,t}{3}\right]-156\,600\,\mathrm{Cos}\left[\frac{4\,t}{3}\right]-362\,000\,\mathrm{Cos}[2\,t]+118\,125\,\mathrm{Cos}\left[\frac{8\,t}{3}\right]+118\,308\,\mathrm{Cos}\left[\frac{10\,t}{3}\right]-52\,500\,\mathrm{Cos}[4\,t]\right)\Big/16\,384\,000\right)+\right.$$
$$\varepsilon^2\left(\frac{231\,131}{262\,144\,000}+\frac{1}{493\,250\,150\,400\,000}\left(-8\,695\,998\,150\,000\,\mathrm{Cos}\left[\frac{2\,t}{3}\right]-27\,990\,084\,188\,080\,\mathrm{Cos}\left[\frac{4\,t}{3}\right]+6\,028\,039\,638\,880\,\mathrm{Cos}[2\,t]-107\,847\,775\,000\,\mathrm{Cos}\left[\frac{8\,t}{3}\right]-\right.\right.$$
$$\left.\left.\left.981\,758\,301\,888\,\mathrm{Cos}\left[\frac{10\,t}{3}\right]-205\,172\,800\,000\,\mathrm{Cos}[4\,t]-402\,871\,967\,280\,\mathrm{Cos}\left[\frac{14\,t}{3}\right]+398\,191\,759\,805\,\mathrm{Cos}\left[\frac{16\,t}{3}\right]-88\,788\,000\,000\,\mathrm{Cos}[6\,t]\right)\right)\right)$$

In[ ]:=

```
Plot[{rr[t] /. ε → 1/2}, {t, 0, 50}]
```

Out[ ]=



Using the solutions, we can define initial conditions and find the corresponding numerical solution for the original equations of motion.

In[ ]:=

```
ωω = ω1 /. rulω /. solω11 /. solω12 /. solω13 /. solr00 /. solr10 /. solr20
```

Out[ ]=

$$1+\frac{\varepsilon}{128}-\frac{129\,\varepsilon^2}{163\,840}-\frac{161\,956\,527\,\varepsilon^3}{41\,943\,040\,000}$$

In[ ]:=

```
initial0 = ({φ[0], ψ[0], φ'[0], ψ'[0], r[t], r'[t]} == {φφ[ωω t], ψψ[ωω t], D[φφ[ωω t], t], D[ψψ[ωω t], t], rr[ωω t], D[rr[ωω t], t]} /. ε → 1/5 /. t → 0) // Thread
```

Out[ ]=

$$\left\{\varphi[0]==0,\ \psi[0]==0,\ \varphi'[0]==\frac{5\,250\,744\,923\,473}{5\,242\,880\,000\,000\,\sqrt{5}},\ \psi'[0]==6\,100\,076\,075\,462\,887\,014\,033\,632\,875\,763\Big/\left(9\,143\,538\,696\,585\,216\,000\,000\,000\,000\,000\,\sqrt{5}\right),\ r[0]==\frac{20\,956\,448\,178\,976\,679}{20\,552\,089\,600\,000\,000},\ r'[0]==0\right\}$$

In[ ]:=

```
sol2a = NDSolve[Join[{eq2 /. k → 13/4}, initial0], {φ[t], ψ[t], r[t]}, {t, 0, 100}]
```

Out[ ]=

```
{{φ[t] → InterpolatingFunction[ Domain: {{0., 100.}} Output: scalar ][t], ψ[t] → InterpolatingFunction[ Domain: {{0., 100.}} Output: scalar ][t], r[t] → InterpolatingFunction[ Domain: {{0., 100.}} Output: scalar ][t]}}
```

In[ ]:=

```
Plot[{φφ[ωω t] /. ε → 1/5, φ[t] /. sol2a}, {t, 0, 80}, PlotStyle → {{Red}, {Thick, Dashed, Blue}}]
```

Out[ ]=



In[ ]:=

```
Plot[{ψψ[ωω t] /. ε → 1/5, ψ[t] /. sol2a}, {t, 0, 50}, PlotStyle → {{Red}, {Thick, Dashed, Blue}}]
```

Out[ ]=



In[ ]:=

```
Plot[{rr[ωω t] /. ε → 1/5, r[t] /. sol2a}, {t, 0, 100}, PlotStyle → {{Red}, {Thick, Dashed, Blue}}]
```

Out[ ]=



We can see that numerical and analytical solutions coincide.

Conclusions

In the present paper we have analyzed an influence of oscillation on the motion of Atwood's machine in the case when both bodies are permitted to oscillate in a plane. We have shown that such oscillations can completely modify the motion of the system. For example, in case of equal masses of the bodies there may exist periodic motion of the system that is very difficult to predict by intuition.

It should be noted that there are many physical problems which can be formulated on the basis of physical laws studied in a standard university course of physics but mathematical models for such problems are quite complicated to be solved and analyzed by hand. Due to this reason such problems are usually not considered in the course of physics. But applying such modern and powerful software as the system *Mathematica* helps a lot in analyzing such problems and promotes development of physical intuition and better understanding of the subject.

References

[1] N.B. Tufillaro, T.A. Abbott, D.J. Griffiths. *Swinging Atwood's machine*. Amer. J. Phys., Vol. 52 (1984) 895–903.

[2] N.B. Tufillaro. *Motions of a swinging Atwood's machine*. J. Physique, Vol. 46 (1985) 1495–1500.

[3] N.B. Tufillaro. *Integrable motion of a swinging Atwood's machine*. Amer. J. Phys., Vol. 54 (1986) 142–143.

[4] N.B. Tufillaro, T.A. Nunes, J. Casasayas. *Unbounded orbits of a swinging Atwood's machine*. Amer. J. Phys., Vol. 56 (1988) 1117–1120.

[5] J. Casasayas, T.A. Nunes, N.B. Tufillaro. *Swinging Atwood's machine: integrability and dynamics*. J. Phys., Vol. 51 (1990) 1693–1702.

[6] H.M. Yehia. *On the integrability of the motion of a heavy particle on a tilted cone and the swinging Atwood machine*. Mech. R. Comm., Vol. 33, No. 5 (2006) 711–716.

[7] O. Pujol, J.P. Perez, J.P. Ramis, C. Simo, S. Simon, J.A. Weil. *Swinging Atwood machine: experimental and theoretical results, and a theoretical study*. Physica D, 239, No. 12 (2010) 1067–1081.

[8] A.N. Prokopenya. *Motion of a swinging Atwood's machine: simulation and analysis with Mathematica*. Mathematics in Computer Science **11**, 417–425 (2017). DOI 10.1007/s11786-017-0301-9

[9] A.N. Prokopenya. *Modelling Atwood's machine with Three Degrees of Freedom*. Mathematics in Computer Science **13**, 247-257 (2019) DOI 10.1007/s11786-018-0357-1

[10] A.N. Prokopenya. *Construction of a periodic solution to the equations of motion of generalized Atwood's machine using computer algebra*. Programming and Computer Software, **46**(2), 120–125 (2020). DOI 10.1134/S0361768820020085

[11] A.N. Prokopenya. *Searching for equilibrium states of Atwood's machine with two oscillating bodies by means of computer algebra*. Programming and Computer Software **47**(1), 43–49. (2021) DOI 10.1007/s11786-018-0357-1